# Copyright 2009 Ryan Phillips
#
# Licensed under the Apache License, Version 2.0 (the "License");
# you may not use this file except in compliance with the License.
# You may obtain a copy of the License at
#
#     http://www.apache.org/licenses/LICENSE-2.0
#
# Unless required by applicable law or agreed to in writing, software
# distributed under the License is distributed on an "AS IS" BASIS,
# WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
# See the License for the specific language governing permissions and
# limitations under the License.

EnsureSConsVersion(1, 2, 0)

opts = Variables('build.py')

opts.Add(PathVariable('APR', 'Path APR','/usr/local/apr/current/bin/apr-1-config'))
opts.Add(PathVariable('APRUTIL', 'Path APR-UTIL','/usr/local/apr/current/bin/apu-1-config'))

protobuf_builder = Builder(name = 'Protobuf', action = 'protoc --cpp_out=. $SOURCES', src_suffix='.proto')
env = Environment(BUILDERS = { 'ProtobufBuilder' : protobuf_builder}, options=opts)

env.ParseConfig(env['APR'] + ' --cflags --cppflags --includes --ldflags --link-ld')
env.ParseConfig(env['APRUTIL'] + ' --includes  --ldflags --link-ld')

env.AppendUnique(CPPPATH = ['#'])
env.AppendUnique(CPPPATH = ['#includes'])
env.AppendUnique(CPPPATH = ['#protocol'])
env.AppendUnique(LIB='protobuf')

conf = env.Configure()

if env['PLATFORM'] == 'darwin':
    env.AppendUnique(LINKFLAGS = ['-undefined', 'dynamic_lookup'])

if not conf.CheckLibWithHeader('protobuf', 'google/protobuf/text_format.h', 'C++'):
    print "Couldn't find protobuf"
    Exit(1)

conf.Finish()

Export('env')
SConscript('protocol/SConscript')
SConscript('client/SConscript')
SConscript('server/SConscript')
